Economic evaluation in clinical trials. You can do it with SAS.

Artur Usov, OCS Consulting, 's-Hertogenbosch, the Netherlands

ABSTRACT
Increasing budget cuts, growing costs and inefficient deployment of current resources in the health care sector increase the necessity for economic evaluation of products brought to market. Therefore the need for Health Economics increases. The aim of this paper is to familiarise the reader with the existing economic evaluation methods in the health care and pharmaceutical industry and to demonstrate the application of the most frequently used evaluation method in clinical trial using SAS 9.3®. The examples provided are based on a fictional clinical trial from which all data is stored in the existing Study Data Tabulation Model (SDTM) domains. Finally, a sensitivity analysis is performed on the findings of the economic evaluations.

INTRODUCTION
The health care expenditures are dramatically increasing over time throughout the world. According to the Organization for Economic Co-operation and Development (OECD), the health care expenditures of developed countries as percentage of Gross Domestic Product have almost doubled over the past 30 years. The World Health Organization identified the following determinants of growing health care expenditures: increase in income, aging population, technological progress and type of health care system implemented in those countries. Technological progress is deemed to be the major driver of health care expenditures.

To counter the trend of growing expenditures, countries around the world have started to base health care utilisation on the evaluation of the cost of new therapies. The values of these costs can be derived from Phase II and III of a clinical trial.

This paper will describe the implementation of a Cost Effectiveness Analysis using SAS 9.3 on a fictional clinical trial. This fictional trial consists of 200 patients. Half of these patients are prescribed with a new treatment, “New Drug”, and the other half with the best alternative, “Old Drug”. This paper will introduce all the components of economics evaluation and will demonstrate how to derive the necessary data for the economic evaluation, perform the evaluation and perform a sensitivity analysis on the findings.

DESIGN OF THE FICTIONAL TRIAL
To demonstrate an economic evaluation a fictional clinical trial is used. The trial consists of 200 patients and two treatment arms “New Drug” and “Old Drug”: “New Drug” is more expensive than “Old Drug”. After each patient has received a treatment he can get hospitalised in one of three types of hospitals: “Regular Hospital”, “Advanced Hospital” and “Emergency Hospital”. Each hospital charges different day rates. Once hospitalised a patient can be prescribed one medication and/or undergo one procedure. The medication and the procedure prices do not vary across hospitals. Also, depending on the condition, a patient can get admitted into a special unit in the hospital. There are two special unit types in the each hospital: a “Regular Unit” and an “Emergency Unit”. Each unit charges different daily rates but those rates do not differ between hospitals.

INCREMENTAL COST EFFECTIVENESS RATIO
There are three methods to evaluate the costs of the therapy: Cost Utility Analysis, Cost Effectiveness Analysis and Cost Benefit Analysis. All three methods are focused on comparing the costs of the therapy versus its effects.

The measurement of costs is the same in all three methods: it would include monetary values of all therapy related medical resources which were consumed. Costs can also include societal costs, like patients who lost time at work or time of informal care provided by relatives. If societal costs are included in the costs measurement, then the societal perspective is taken. On the other hand if only medical resources are included then we are dealing with the payers perspective. In this paper we will focus on the payers perspective, thus will include only medical resources.

The measurement of effects differ between the analyses methods. In cost benefit analysis, all effects (reduced number of hospital days, lower mortality rates, etc) are derived in monetary terms. In cost effectiveness analysis all effects are measured in natural units, e.g.: reduced sugar levels or blood pressure. Those methods will not be
covered in this paper. Instead, we focus on Cost Utility analysis where effects are measured in Quality Adjusted Life Years.

Cost Utility Analysis is designed to calculate the ratio of incremental costs versus the incremental effects of a new therapy versus the best available alternative. This ratio is called Incremental Cost Effectiveness Ratio (ICER) and its mathematical representation can be found below.

\[
ICER = \frac{\text{Costs}_{\text{treat}} - \text{Costs}_{\text{control}}}{\text{Effects}_{\text{treat}} - \text{Effects}_{\text{control}}} = \frac{\Delta C}{\Delta Q}
\]

In the formula above, Costs_{treat} and Effects_{treat} are costs and effects of the treatment group and Costs_{control} and Effects_{control} are costs and effects of the comparator (control group). By finding the difference in costs and effects between the treatment group and the comparator we get a ratio of Incremental Costs (\Delta C) divided by Incremental Effects (\Delta E). That ratio shows how much a change in effect would cost. Each component of the ratio will be described in detail in the next sections.

Depending on the costs and effects data, four possible ICER scenarios are possible:

1) New therapy is more expensive and more effective (A)
2) New therapy is more expensive and less effective (B)
3) New therapy is less expensive and less effective (C)
4) New therapy is less expensive and more effective (D)

All four scenarios can be put on a cost – effectiveness quadrant (Figure 1) with incremental costs on the y-axis and incremental effects on the x-axis.

The position of the ICER in the quadrant can help decide if the new therapy justifies the cost. If the new therapy is represented by a scenario D, then the decision can be made in favor of the new therapy as it saves money and brings gain in health. The opposite is true for scenario B. Scenarios A and C are not as straightforward. In scenario A gain in health comes with increased costs. The opposite holds for scenario C. In those cases implementation of the therapy will depend on whether the health gain (or costs saving) is worth the additional costs (or health loss).

COSTS
Measuring the costs of the therapy is done in three stages: 1) identify the categories of the resource likely to be used 2) measure how much of each resource is required 3) value the used resources by applying unit costs to each resource group. As stated earlier, identification of used resources depends on the perspective of the analysis. Two mostly used perspectives in the analyses are societal and payer perspectives. The societal perspective not only includes the medical resource of the therapy but also includes other costs such as loss of time by friends for providing informal health care, lost days at work, etc. In this paper we focus on payer perspective thus only health care related resources are measured.

Use of resources can be pooled from following Study Data Tabulation Model (SDTM) domains: Concomitant Medications (CM), Hospitalizations (HO), Adverse Events (AE) etc. The medical resource data for each patient can later be linked onto the prices data and pooled together.

In our fictional clinical trial we derive used resources from the following events: cost of the treatment, length of stay in the hospital, type of hospital where the patient staid, use of special unit in the hospital and use of medication/procedures during the stay in the hospital. All used data on resource usage, apart from treatment cost, is taken from HO and SUPPHO SDTM domains.
An example of the data in the HO domain is shown in Figure 2. Here: HOSTDTC and HOENDTC are the start/end date of hospitalisation, USUBJID is the unique subject identifier, HODUR is the duration of the hospitalization in days and the remaining variables are standard SDTM variables.

![Table 2: Example of hospitalization data in HO domain]

An example of the data entry in the SUPPHO domain is shown in Figure 3. SUPPHO holds detailed information on the hospitalisation: if an adverse event was reported during this hospitalisation, if any medication were prescribed, if any procedures were performed, the name of the hospital, the type of special unit in the specified hospital and if special unit was used during hospitalization.

![Table 3: Example of additional hospitalization data in SUPPHO domain]

Tables 1 indicate the cost for each resource group. Each patient can get into one of the hospitals based on the severity of their condition. A regular hospital charges 1000 euros per day, an advanced hospital 2000 euros per day and an emergency hospital 3000 euros per day. Within each hospital a patient can be brought to a regular or an emergency special unit. The daily rate of a regular unit is 400 euros and an emergency unit 750 euros. In our fictional trial we assume that if a patient is assigned to a special care unit he spends the entire hospital stay there. Special unit types do not vary between hospitals. The same holds true for prescribed medication and procedure costs which are equal to 712 and 2015 euros per admission/prescription.

<table>
<thead>
<tr>
<th>Hospital</th>
<th>Regular Hospital</th>
<th>Advanced Hospital</th>
<th>Emergency Hospital</th>
</tr>
</thead>
<tbody>
<tr>
<td>Hospital Costs</td>
<td>1000</td>
<td>2000</td>
<td>3000</td>
</tr>
<tr>
<td>Medication Costs</td>
<td>712</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Procedure Costs</td>
<td>2015</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

![Table 1: Costs for resource groups]

Before linking the costs and the resource data together, the resource data from the SUPPHO domain has to be transposed. Later it will be linked with the HO domain to obtain the duration of the hospitalisation data. The following SAS code transposes the HO domain and links the costs and resource use data together:

```sas
proc transpose data=work.suppho
   out =work.suppho2(rename=(idvarval=hoseq));
var qval;
id qnam;
by studyid usubjid idvarval;
run;

data work.costs1;
set work.suppho2;
/*Costs per type of hospital*/
if provnm="Regular Hospital" then t_hosp_costs=1000;
else if provnm="Advanced Hospital" then t_hosp_costs=2000;
else if provnm="Emergency Hospital" then t_hosp_costs=3000;
/*Costs for special units. */
if spunfl="Y" then do;
   if spuncd="Regular Unit" then t_hosp_costs=sum(t_hosp_costs,400);
```

3
else if spuncd="Emergency Unit" then t_hosp_costs=sum(t_hosp_costs,750);
end;

/*Costs for medications and procedures*/
if procfl="Y" then proc_costs=2015;
if medsfl="Y" then med_costs=712;
keep studyid usubjid hoseq t_hosp_costs proc_costs med_costs;
run;

Once the cost data is linked, the total costs per hospitalisation can be calculated as duration in the hospital, multiplied by the hospital costs per day plus the medication and procedure costs. The hospitalisation duration data is taken from the HO domain.

proc sql;
/*Calculating the hospitalization costs per hospitalization*/
create table work.total_costs1 as
select b.*,
    sum(hodur*t_hosp_costs,proc_costs,med_costs) as hosp_costs
from work.ho as a right join work.costs3 as b
on a.studyid=b.studyid and a.usubjid=b.usubjid and a.hoseq=b.hoseq;
/*Calculating the total costs per patient*/
create table work.total_costs2 as
select usubjid,
    sum(hosp_costs) as pat_costs
from work.total_costs1
group by usubjid;
quit;

Figure 4 shows the sample cost data. In this paper costs are not discounted. To get familiar with discounting of the costs you can read the articles mentioned in the “recommended reading” section.

<table>
<thead>
<tr>
<th>USUBJID</th>
<th>PAT_COSTS</th>
</tr>
</thead>
<tbody>
<tr>
<td>2</td>
<td>7392</td>
</tr>
</tbody>
</table>

Figure 4: Example of total costs data

EFFECTS
The second component of the ICER is effects. As mentioned earlier, in this paper we use Cost Utility Analysis and effects are measured in Quality Adjusted Life Years (QALY).

Quality Adjusted Life Years (QALY) is a generic measure of years lived after adjusting for the quality of life during that period. Depending on the health state a weight (utility) is assigned for each period of life. Weight can vary from 1 to 0, where 1 is assigned to perfect health and 0 to death. To obtain total QALY per patient we summarise the assigned weights for all periods. Consider the following example: If a patient lived one year in perfect health the corresponding QALY is 1 (1 year*1 full utility). In the second year, if the utility of that individual has reduced by half due to his illness then in the second year that individual will have 0.5 QALYs (1year*0.5 utility). Thus in the course of two years that individual had enjoyed 1.5 QALYs.

In this paper we are going to use the EQ-5D-3L questionnaire to measure utilities assigned for each period, but other questionnaires are available and can be found in the recommended reading section.

The EQ-5D-3L asks a patient to rate his current state in 5 dimensions: Mobility, Self-Care, Usual-Activities, Pain/Discomfort, Anxiety/Depression. Each dimension has 3 levels, where 1 is the best state and 3 is the worst. A sample of the EQ-5D-3L questionnaire is presented in Figure 5.
Once the scores for all 5 dimensions are filled, a health state can be presented as a 5-digit number. A 11111 state would indicate perfect health and would have a utility equal to 1. Any deviation from that state would be interpreted as decrement from full health. For example: if a patient selected 1 for Mobility, 1 for Self-Care, 1 for Usual Activities, 1 for Pain/Discomfort and 2 for Anxiety/Depression then the state is represented as 11112. In this state the patient has no problems with mobility, self-care, usual activities or pain, but has a moderate depression. Since there is a deviation from perfect health a certain coefficient has to be deducted from 1 (utility of perfect health). Also, additional constant coefficients are deducted for any deviation from perfect health and, if a 3 (worse state) was selected, at least once in any dimension. All the coefficients represent the general opinion of the public on than state. Deviation coefficients differ between the five dimensions. Also, those coefficients differ between counties.

In our fictional trial the utility scores, for each level of each dimension, were downloaded from “The Economics Network” website and were imported to SAS. The coefficients for deviations can be seen in Figure 6. Next to these figures, the coefficient for any deviation from perfect health is set to 0.071 and to 0.234 if a patient indicated 3 in any of five dimensions. (These values were also obtained from the “The Economics Network” website). For the state 11112 a utility score would be calculated as 1 – 0.124 - 0.071 = 0.805, where 0.124 is a coefficient for deviation from no depression and 0.071 is a coefficient for any deviation from perfect health.

<table>
<thead>
<tr>
<th>Scope</th>
<th>Level</th>
<th>Coefficient</th>
</tr>
</thead>
<tbody>
<tr>
<td>Mobility</td>
<td>2</td>
<td>0.036</td>
</tr>
<tr>
<td>Mobility</td>
<td>3</td>
<td>0.161</td>
</tr>
<tr>
<td>Self-Care</td>
<td>2</td>
<td>0.082</td>
</tr>
<tr>
<td>Self-Care</td>
<td>3</td>
<td>0.152</td>
</tr>
<tr>
<td>Usual Activities</td>
<td>2</td>
<td>0.032</td>
</tr>
<tr>
<td>Usual Activities</td>
<td>3</td>
<td>0.057</td>
</tr>
<tr>
<td>Pain/Discomfort</td>
<td>2</td>
<td>0.086</td>
</tr>
<tr>
<td>Pain/Discomfort</td>
<td>3</td>
<td>0.329</td>
</tr>
<tr>
<td>Anxiety/Depression</td>
<td>2</td>
<td>0.124</td>
</tr>
<tr>
<td>Anxiety/Depression</td>
<td>3</td>
<td>0.325</td>
</tr>
</tbody>
</table>

Figure 6: EQ-5D-3L coefficients for deviation from perfect health

All data is stored in the QC SDTM domain. The date of the collection is stored in the variable QSDTC, short names of dimensions are stored in the variable QSTESTCD, full name of the dimensions are stored in the variable QSCAT, coded answer of the patient for each dimension are stored in the variable QSORRES and the verbal answer is stored in the variable QSORRESC. An example of this data is shown in Figure 7.
The coefficients data and questionnaire data are linked by using the QSCAT/Scope and QSCORRES/Level variables. The following code was used to calculate the utility scores for each individual and the resulting sample of data is presented in Figure 8. In our fictional trial QALY, the data is not discounted. To get familiar with discounting of the QALYs you can read the articles mentioned in the “recommended reading” section.

```ulaql
data work.trial_qaly2;
    keep STUDYID USUBJID QSBFL QSDTC QALY;
    set work.trial_qaly1;
    /*Variable state will hold the numeric representation of the health state*/
    retain state qaly;
    length state $10;
    by studyid USUBJID QSDTC;
    if first.qsdtc then do;
        call missing(state);
        /*Setting the health state to perfect health. If any deviations will be present coefficients will be deducted from one.*/
        qaly=1;
    end;
    /*Deducting points from the perfect state is coefficient is not missing. It will be missing for no deviations from perfect level of mobility, self-care, usual activities pain and anxiety.*/
    if not missing (coefficient) then qaly=qaly-coefficient;
    /*Recording the health state in a five digit number*/
    state=catx("," ,state,put(QSORRES,best12.));
    if last.qsdtc then do;
        /*Deducting additional points if there was a deviation. */
        if find(state,"2")>0 or find(state,"3")>0 then do;
            qaly=qaly-utility;
            qaly=qaly-0.071;
        end;
        if find(state,"3")>0 then qaly=qaly-0.234;
        output;
    end;
run;
```

![Figure 7: Example of EQ-5D-3L questionnaire data in the QS domain](image)

![Figure 8: Example QALY data obtained from the EQ-5D-3L questionnaire](image)

**ANALYSIS**

Now that both the effects and the costs are collected, it is possible to perform the Cost Utility Analysis.

First we merge the costs and effect data together and add treatment name from the DM domain. Next, we add the treatment costs to the costs variable (“New Drug” costs 1500 Euros and the “Old Drug” costs 2010 Euros).
By using the means procedure we can calculate the arithmetic mean costs and effects per treatment.

```sas
proc means data=work.costs_eff_1 noprint;
  var pat_costs;
  by armcd;
  output out=work.stats_costs1 mean=mean_cost std=std_cost stderr=stderr_cost;
run;

proc means data=work.costs_eff_1 noprint;
  var qualy_disc;
  by armcd;
  output out=work.stats_qaly1 mean=mean_effect std=std_effect stderr=stderr_effect;
run;
```

![Figure 9: Summary statistics of effects](image1.png)

<table>
<thead>
<tr>
<th>ARMC</th>
<th>mean_effect</th>
<th>std_effect</th>
<th>stderr_effect</th>
</tr>
</thead>
<tbody>
<tr>
<td>New Drug</td>
<td>3.00267</td>
<td>0.1772526245</td>
<td>0.0177252847</td>
</tr>
<tr>
<td>Old Drug</td>
<td>0.97452</td>
<td>0.4773708611</td>
<td>0.0477370861</td>
</tr>
</tbody>
</table>

![Figure 10: Summary statistics of costs](image2.png)

<table>
<thead>
<tr>
<th>ARMC</th>
<th>mean_cost</th>
<th>std_cost</th>
<th>stderr_cost</th>
</tr>
</thead>
<tbody>
<tr>
<td>New Drug</td>
<td>3820.74</td>
<td>1970.4482678</td>
<td>197.0448268</td>
</tr>
<tr>
<td>Old Drug</td>
<td>5231.46</td>
<td>2239.9963313</td>
<td>223.9963313</td>
</tr>
</tbody>
</table>

ICER is a measure which is obtained by dividing the incremental costs of the new treatment by the incremental effects of the new treatment. To calculate the ICER in SAS, first the results from the means procedure have to be merged and transposed for each treatment. Then, by using a retain statement the difference in costs and effects between the new and the old treatment is calculated. Once the incremental values are obtained, the ICER is calculated. The code below reflects the above mentioned steps.

```sas
/*Merging mean effect and costs per treatment*/
proc sql;
  create table work.icer1 as
  select a.armcd,
    a.mean_cost,
    b.mean_effect
  from work.stats_costs1 as a
    left join work.stats_qaly1 as b
  on a.ARMCD = b.ARMCD;
quit;

/*Descending option is used in order to have first the Old drug and then the new drug. The difference will be calculated as old-new.*/
proc sort data=work.icer1;
  by descending armcd;
run;

/*Transposing the data*/
proc transpose data=work.icer1
  out=work.icer2(rename=(col1=result));
  by descending armcd;
run;

data work.icer3(keep=measure difference varname);
  /*Using a retain statement to calculate the difference in effects and costs*/
  retain cost effect;
  set work.icer2;
  length measure $30;
  /*Initialising the values of effect and costs with values of the first treatment*/
  if armcd="Old Drug" then do;
    if _name_="mean_cost" then cost=result;
    if _name_="mean_effect" then effect=result;
  end;
```

Figure 9: Summary statistics of effects
Figure 10: Summary statistics of costs
Finding the difference in the effects are costs and outputting those values:

```plaintext
/*Finding the difference in the effects are costs and outputting those values*/
if armcd='New Drug' then do;
    if _name_='mean_cost' then do;
        measure='Difference in costs';
        varname='COSTDIF';
        difference=result-cost;
        output;
    end;
    if _name_='mean_effect' then do;
        difference=result-effect;
        measure='Difference in effects';
        varname='EFETDIF';
        output;
    end;
end;
run;

/*Transposing the incremental costs and effect data to have one record*/
proc transpose data=work.icer3
    out=work.icer4
    label=measure;
    id varname;
run;
/*Calculating ICER*/
data work.icer_final;
    set work.icer4;
    keep icer;
    icer=costdif/efetdif;
run;
```

<table>
<thead>
<tr>
<th>COSTDIF</th>
<th>EFETDIF</th>
<th>ICER</th>
</tr>
</thead>
<tbody>
<tr>
<td>-1.46072</td>
<td>2.02815</td>
<td>-720.2228632</td>
</tr>
</tbody>
</table>

Figure 11: Estimated incremental costs, incremental effects and ICER

The results in Figure 11 show that “New Drug” has lower costs and higher effects compared to the “Old Drug”. The resulting ICER is equal to -720 Euros/QALY. If we place the resulting ICER on the cost-effectiveness quadrant presented in the “INCREMENTAL COST EFFECTIVENESS” section it would appear in the part D: new therapy has lower costs and higher effects. In that scenario the treatment drug dominates the comparator.

SENSITIVITY ANALYSIS
Since clinical trials include only a sample of the population instead of the entire population, decision makers require a sensitivity analysis of the findings. Sensitivity analysis shows how sensitive the findings are in case any of the parameters is changed. In this paper we will produce the confidence intervals for the ICER and create the Acceptability Curve.

CONFIDENCE INTERVALS
Since ICER is a ratio, it is inappropriate to use standard statistical techniques to construct confidence intervals. Instead, other techniques like non-parametric bootstrapping can be used. In the non-parametric bootstrap, subsamples are redrawn from the original sample N times. So for a sample of 100 patients we can create N samples with 100 patients, which are drawn from the original sample. Please note that patients can occur more than once.

Once the bootstrap samples are created, the ICER is calculated for each observation in the dataset.
PhUSE 2013

```sas
end;
end;
stop;
run;

data bootsamp2;
  do sampnum = 1 to 10000; /* To create 1000 bootstrap replications New Drug data*/
    do i = 1 to nobs;
      /*Selecting random 100 observation from treat2_3 dataset which has patient level data of costs and effects of new drug*/
      x = round(ranuni(0) * nobs);
      set work.treat2_2
      nobs = nobs
      point = x;
      output;
    end;
  end;
stop;
run;

/*Sorting and merging data of new and old drugs together. The resulting dataset with have paired data of costs and effect per record.*/
proc sort data=work.bootsamp1;
  by i;
run;

proc sort data=work.bootsamp2;
  by i;
run;

data work.bootsamp_comb1;
  merge work.bootsamp1(keep=i pat_costs qualy_disc)
    rename=(pat_costs=pat_costs1 qualy_disc=qualy_disc1)
  work.bootsamp2(keep=i pat_costs qualy_disc)
    rename=(pat_costs=pat_costs2 qualy_disc=qualy_disc2));
  by i;
/*Calculating ICERs, incremental costs and effects for each pair of patients in the dataset*/
  ICER=(qualy_disc2-qualy_disc1)/(pat_costs2-pat_costs1);
  cost_diff=pat_costs2-pat_costs1;
  eff_diff=qualy_disc2-qualy_disc1;
run;

/*Using univariate procedure to calculate the mean value, 2.5th and 97.5th percentiles*/
proc univariate data=work.bootsamp_comb1;
  var icer;
  output out=work.stats_boot_icer
    mean=mean_icer
    pctlpre=P pctlpts= 2.5, 97.5;
run;
```

The calculated 2.5th and 97.5th percentiles in Figure 12 represent the two tailed 95% confidence interval of the ICER. We can also plot the calculated ICER’s to see the distribution. As indicated in Figure 13, most of the values have a positive difference in effect and negative difference in the costs, thus in most cases new therapy dominates the old one.

Figure 12: Mean value of ICER and a 95% confidence interval
Each decision maker has to spend a finite amount of resources for each therapy. But since data from the clinical trial represents only the sample from the population, decision maker cannot be sure about the true costs of the therapy. Luckily a technique called Acceptability Curve is available.

An Acceptability Curve represents the probability that the new treatment will be cost effective for an accepted amount of money the decision maker is willing to pay per additional QALY. The principle of deriving such a graph which is presented in Figure 15, is illustrated in Figure 14. The red line represents the maximum amount of euros a decision maker is willing to pay/save per increase/decrease in effects. That maximum amount is a slope. The black dots in the graph represent the bootstrapped ICER values. In this scenario 4 values are below the red line and 2 values are above the line. In other words, there is a 67% probability that the new treatment will be costs effective for this willingness to pay for value. An acceptability curve can be created by rotating the red line and plotting the probability values of the ICERS being below the red line against the corresponding willingness to pay.

In our fictional study we have created a macro called acc_curve, which calculates the probability of ICERS below the willingness to pay. The willingness to pay is varied from 0 to 4000 in steps of 20.

```sas
%macro acc_curve();
  /*Varying willingness to pay from 0 to 4000 euros with a step of 1000 euros*/
  %do i=0 %to 4000 %by 20;
```

Checking if the calculated ICER is below or above the maximum willingness to pay. If it is below value 1 is prescribed and 0 otherwise.

```sas
data work.acc_curve_&i;
    set work.bootsamp_comb1;
    max_pay=input("&i", best12.);
    if eff_diff<0 then do;
        if icer<max_pay then acc=0;
        else acc=1;
    end;
    else do;
        if icer<max_pay then acc=1;
        else acc=0;
    end;
run;
```

/*Using proc sql to calculate the mean number of ICERs which were below the maximum willingness to pay.*/
```
sql
create table acc_fin_&i as
    select max_pay ,
    mean(acc) as prob
    from work.acc_curve_&i
    group by max_pay;
quit;
```

/*Appending the obtained data to an empty shell dataset*/
```
data work.acc_fin_tot;
    set work.acc_fin_tot acc_fin_&i;
run;
```
```
proc datasets noprint;
    delete  acc_fin_&i
        acc_curve_&i;
quit;
```
```
%end;
%mend acc_curve;
```
```
%acc_curve();
```

**Acceptability Curve**

![Figure 15: Acceptability curve](image)

Figure 15: Acceptability curve
Based on the acceptability curve in Figure 15 we see that as the willingness to pay increases, the probability that the ICER is below the maximum willingness to pay also increases. Somewhere around 3800 euros the probability starts approaching 1. This means that if a decision maker is willing to pay 3800 euros he can be almost 100% sure that the therapy will be cost effective.

**CONCLUSION**  
With rising costs and stricter budgets, pharmaceutical companies have to provide not only efficacy and safety data but also justify the costs of the new therapies. This paper uses very basic economic analysis, but already demonstrates that SAS can be a powerful tool to help visualize the value of new therapies and assist decision makers in making the right decision for the utilisation of the therapies.

**REFERENCES**


**ACKNOWLEDGMENTS**

I would like to thank Raymond Ebben and Jules Van der Zalm for support and guidance. Also, I would like to thank Thodoris Chatzivasiliadis for reviewing my work.

**RECOMMENDED READING**


**CONTACT INFORMATION**

Your comments and questions are valued and encouraged. Contact the author at:

Artur Usov  
OCS Consulting  
P.O. Box 3434  
’s-Hertogenbosch  
+31 (0)73 523 6000  
sasquestions@ocs-consulting.com  
http://www.ocs-consulting.nl/

Brand and product names are trademarks of their respective companies.